Coxeter Groups and Complex Reflection Groups

Don Taylor

The University of Sydney

16 February 2012

Symmetries of Discrete Objects
Queenstown

Part I. Coxeter groups

Definition

A *Coxeter matrix* of rank n is a symmetric $n \times n$ matrix $M = (m_{ij})$ with non-negative integer entries such that $m_{ii} = 1$ for all $i \in [n] = \{1, \ldots, n\}$.

Definition

The *Coxeter group* associated with the Coxeter matrix M is the group $W(M)$ with generators $R = \{r_i \mid i = 1, \ldots, n\}$ and relations

$$(r_i r_j)^{m_{ij}} = 1 \quad \text{for all } i, j \in [n].$$

The pair $(W(M), R)$ is a *Coxeter system*.

It is a theorem that the order of $r_i r_j$ is m_{ij} when $m_{ij} \neq 0$ and that it is infinite when $m_{ij} = 0$. In particular, the generators have order 2.

Coxeter groups in Magma

Magma can represent a Coxeter group in three ways:

1. as a finitely presented group: i.e., by generators and relations, as on the previous slide;
2. as a permutation group;
3. as a matrix group

By the way, typing `CoxeterGroup;` at a Magma prompt produces a large list of signatures. You can cut this down a bit by asking for just those signatures which involve a given category.

```
> ListSignatures(CoxeterGroup,GrpPermCox);
```

Magma code: infinite Coxeter groups

If the group $W(M_1)$ of a Coxeter matrix M_1 is infinite, the intrinsic function `CoxeterGroup(M1)` returns a finitely presented group.

```
> M1 := Matrix(3,3,[1,3,3, 3,1,3, 3,3,1]);
> IsCoxeterMatrix(M1);
true
> W1 := CoxeterGroup(M1);
> IsFinite(W1);
false
> W1;
Coxeter group: Finitely presented group on 3 generators
Relations
    $.1 * $.2 * $.1 = $.2 * $.1 * $.2
    $.1 * $.3 * $.1 = $.3 * $.1 * $.3
    $.2 * $.3 * $.2 = $.3 * $.2 * $.3
    $.1^2 = Id($)
    $.2^2 = Id($)
    $.3^2 = Id($)
```

Magma code: finite Coxeter groups

The Coxeter matrix of a finite Coxeter group can be obtained by giving its 'Cartan name'. For example,

```
> M2 := CoxeterMatrix("A2H3");
> M2;
[1 3 2 2 2]
[3 1 2 2 2]
[2 2 1 5 2]
[2 2 5 1 3]
[2 2 2 3 1]
```

In this case $W(M_2)$ is finite and so `CoxeterGroup(M2)` returns a permutation group. The finitely presented group is obtained by explicitly referring to the category `GrpFPCox`:

```
> FPW2 := CoxeterGroup(GrpFPCox,M2);
```

Magma code: Coxeter groups as matrix groups

The representation of $W(M_2)$ as a matrix group is obtained by specifying `GrpMat` as the category.

```
> MW2 := CoxeterGroup(GrpMat,M2);
> MW2:Minimal;
MatrixGroup(5, Number Field with defining polynomial x^2 - x - 1 over
the Rational Field) of order 2^4 * 3^2 * 5
> CartanName(MW2);
A2 H3
```

Every Coxeter group has a matrix representation but not necessarily a permutation representation in Magma.

```
> MW1 := CoxeterGroup(GrpMat,M1);
> IsFinite(MW1);
false
```

```
> PW1 := CoxeterGroup(GrpPermCox,M1);
>> PW1 := CoxeterGroup(GrpPermCox,M1);
                    ^
Runtime error in 'CoxeterGroup': Only finite groups have a permutation
representation
```

Coxeter diagrams

Definition

The *Coxeter diagram* of a Coxeter group $W(M)$, where M is a Coxeter matrix of rank n is the labelled graph on n vertices $\{v_i \mid i \in [n]\}$ such that for $i \neq j$ there is an edge labelled m_{ij} between v_i and v_j whenever $m_{ij} \geqslant 3$. If $m_{ij} = 0$ the vertices v_i and v_j are joined by an edge labelled ∞. If $m_{ij} = 3$, the label is usually omitted.

```
> CoxeterDiagram(MW2);

A2    1 - 2


H3    3 -- 4 - 5
         5
```

Coxeter introduced these diagrams in 1931 and in 1946 E. B. Dynkin used similar diagrams in his work on semisimple Lie algebras.

Roots and reflections

The usual notion of *reflection* is an orthogonal transformation which fixes every vector of some hyperplane and sends the vectors perpendicular to the hyperplane to their negatives.

That is, if E is a finite dimensional Euclidean space with inner product $(-,-)$, then a reflection in E is a linear transformation r which can be defined by the formula

$$r(v) = v - \frac{2(v,a)}{(a,a)}a$$

for some vector a, called a *root* of r. The reflecting hyperplane is $a^{\perp}$.



Coxeter groups and reflections

Given a Coxeter group $W = W(M)$ defined by a Coxeter matrix of rank n, let V be the real vector space of dimension n with basis $\{e_i \mid i \in [n]\}$ and define an inner product $(-,-)$ on V by

$$(e_i, e_j) = \begin{cases} -1 & m_{ij} = 0, \\ -\cos(\pi/m_{ij}) & m_{ij} \neq 0. \end{cases}$$

For $i \in [n]$, let σ_i be the reflection

$$\sigma_i(v) = v - 2(v, e_i)e_i.$$

The map $r_i \mapsto \sigma_i$ extends to an isomorphism σ from W to the subgroup of $\mathrm{GL}(V)$ generated by the reflections σ_i. The space V is the *reflection representation* of W.

A Coxeter group is finite if and only if the inner product defined above is positive definite; i.e., it is a *Euclidean* reflection group.

The root system

Identify W with its image $\sigma(W)$. Then the vectors e_i are roots of the reflections r_i: they are called the *simple* roots of W.

If $r \in W$ is a reflection and if a_r is a root of r such that $(a_r, a_r) = 1$, we may choose a_r so that it is a *non-negative* linear combination of simple roots. The set Φ^+ of the a_r is the set of *positive roots* of W. The *root system* is $\Phi = \Phi^+ \cup (-\Phi^+)$.

```
> W := CoxeterGroup("E7");
> Phi := RootSystem(W);
> NumPosRoots(Phi);
63
```

The permutation representation of W used by MAGMA is on the set of roots.

```
> W;
Coxeter group: Permutation group W acting on a set of cardinality 126
Order = 2903040 = 2^10 * 3^4 * 5 * 7
```

A graph on 27 vertices

```
> E7 := CoxeterGroup("E7");
> R := RootSystem(E7);
> J := CoxeterForm(R);
> V := VectorSpace(Rationals(),7,J);
> P := PositiveRoots(R);
> ChangeUniverse(~P,V);
> X := { v : v in P | v[7] eq 1};
> #X;
27
> A := Matrix(27,27,[2*InnerProduct(u,v) : u,v in X])
>       - 2*IdentityMatrix(Integers(),27);
> gr := Graph< 27 | A >;
> G := AutomorphismGroup(gr);
> #G;
51840
> flag, _ := IsIsomorphic(G,CoxeterGroup("E6"));
> flag;
true
```

Weyl groups, Lie algebras, reductive algebraic groups

Definition

A *Weyl group* is a Coxeter group W whose reflection representation has a W-invariant $\mathbb{Z}$-lattice. Therefore its elements can be represented by matrices with integer entries.

If W is a Weyl group it is customary to scale the roots so that every root is an *integer* linear combination of simple roots.

Weyl groups are associated with semisimple Lie algebras and with reductive algebraic groups.

In order to describe the correspondence between Weyl groups and groups of Lie type we need the concept of a *root datum*.

The correspondence is between reductive algebraic groups and pairs (R, F), where R is a root datum with Weyl group W, and F is a field.

Root data

A *root datum* is a 4-tuple (X, Φ, Y, Φ^*) such that:

- X and Y are free $\mathbb{Z}$-modules of finite rank in duality via a pairing $(x, y) \mapsto \langle x, y \rangle$;
- Φ is a finite subset of X and Φ^* is a finite subset of Y;
- there is a bijection $\Phi \to \Phi^* : \alpha \mapsto \alpha^*$ such that $\langle \alpha, \alpha^* \rangle = 2$;
- for $\alpha \in \Phi$ the map $r_\alpha(x) = x - \langle x, \alpha^* \rangle \alpha$ is a reflection which preserves Φ;
- for $\alpha \in \Phi$ the map $r^*_\alpha(y) = y - \langle \alpha, y \rangle \alpha^*$ is a reflection which preserves Φ^*.

Given a root datum Σ, for every field F there is a unique connected reductive group $G(F)$ with a maximal torus T such that Σ is the root datum of $G(F)$ with respect to T. (The reflections r_α generate the Weyl group of $G(F)$.)

Hecke algebras

Let (W, R) be a Coxeter system and let $L = \mathbb{Z}[q^{-1}, q]$ be the ring of Laurent polynomials.

Definition

The *Hecke algebra* $\mathcal{H}$ of (W, R) is the associative L-algebra generated by elements $\{T_r \mid r \in R\}$ subject to the relations

$$T_r^2 = 1 + (q - q^{-1})T_r$$
$$\underbrace{T_rT_sT_r\cdots}_{m_{rs}} = \underbrace{T_sT_rT_s\cdots}_{m_{rs}}$$

where m_{rs} is the order of rs.

W-graphs

Let (W, R) be a Coxeter system.

Definition

A *W-graph* is a (directed or undirected) graph with vertex labels and edge weights. The label attached to a vertex v is a subset of R (called the *descent set* of v) and the edge weights are scalars (usually integers). If Γ is the vertex set, E the edge set, and $\mu : E \to \mathbb{Z}$ the edge weights, the free L-module $L\Gamma$ should be an $\mathcal{H}$-module with respect to the action

$$uT_r = \begin{cases} -q^{-1}u & u \in I(u), \\ qu + \sum_{\{v \in V \mid s \in I(v)\}} \mu(v, u)v & s \notin I(u). \end{cases}$$

Sparse matrix representations

```
> C6 := SymmetricMatrix(
>   [1, 3,1, 2,3,1, 2,3,2,1, 2,2,2,3,1, 2,2,3,2,2,1]);
> E6 := CoxeterGroup(GrpFPCox,C6);

> specht32, A4 := Partition2WGtable([3,2]); // dimension 5
> table := InduceWGtable([1,2,4,5],specht32,E6);
> wg := WGtable2WG(table);

> Hreps := WG2HeckeRep(E6,wg);
> #Hreps;
6
> Hreps[1];
Sparse matrix with 2160 rows and 2160 columns over
Univariate rational function field over Integer Ring
> Greps := WG2GroupRep(E6,wg);
> Greps[1];
Sparse matrix with 2160 rows and 2160 columns over Integer Ring
```

Part II. Pseudo-reflections

Let V be a vector space of dimension n.

A *pseudo-reflection* [Bourbaki] is a linear transformation of V whose space of fixed points is a hyperplane. This includes transvections and projections as well as reflections in Euclidean and Hilbert spaces.

If r is a pseudo-reflection, a *root* of r is vector a which spans $\mathrm{Im}(1-r)$. Then there exists $\varphi \in V^*$ such that $r = r_{a,\varphi}$, where

$$vr_{a,\varphi} = v - \varphi(v)a.$$

The linear functional φ is the *dual root* of a and $\ker\varphi$ is the hyperplane of fixed points of r.

If $\varphi(a) \neq 0, 1$ the pseudo-reflection $r_{a,\varphi}$ is a *reflection*.

We have $r(a) = (1-\varphi(a))a$ and therefore, if the order of r is finite, $\zeta = 1 - \varphi(a)$ is a root of unity.

Unitary reflection groups

A *complex reflection group* is a group G generated by a (finite) number of reflections (of finite order) of a vector space V over $\mathbb{C}$.

If G is finite, then G preserves a positive definite hermitian form $(-,-)$ and so G is generated by *unitary* reflections, where the action of a unitary reflection r_a with root a and eigenvalue ζ is given by

$$vr_a = v - (1-\zeta)\frac{(v,a)}{(a,a)}a.$$

In this case we say that G is a *unitary reflection group*.

The vector $\frac{(1-\overline{\zeta})}{(a,a)}a$ is the *coroot* of a.

Example: the group G_6

The Shephard and Todd group $W = G_6$ is a primitive complex reflection group of rank 2. It is isomorphic to $\mathbb{Z}_4 \circ \mathrm{SL}_2(\mathbb{F}_3)$ and it is generated by a reflection r or order 2 and a reflection s of order 3:

$$r = \begin{pmatrix} -1 & 0 \\ -i(1+\sqrt{3}) & 1 \end{pmatrix} \quad \text{and} \quad s = \begin{pmatrix} 1 & -\omega^2 \\ 0 & \omega \end{pmatrix}$$

There are 6 conjugates of r and 4 conjugates of s. The group is defined over the field $K = \mathbb{Q}[i, \omega]$, where $i^2 = -1$ and $\omega^3 = 1$.



Complex reflection groups in MAGMA

The function `ShephardTodd(n)` returns the *primitive* Shephard and Todd group G_n.

The generating matrices are written over the ring of integers of the minimal field of G_n, which is the field generated by the character values of the reflection representation.

By default the ambient field of G_n in MAGMA is a cyclotomic field, which in some cases is larger than the minimal field. If you wish you can instruct MAGMA to use a number field (with slower arithmetic operations).

The function `ShephardTodd(m,p,n)` returns the *imprimitive* complex reflection group $G(m, p, n)$.

Magma code

```
> G := ShephardTodd(6);
> G;
MatrixGroup(2, Cyclotomic Field of order 12 and degree 4)
Generators:
    [                 -1                    0]
    [-z^3 - 2*z^2 + 1                       1]

    [       1      z^2]
    [       0 z^2 - 1]
> G := ShephardTodd(6 : NumFld);
> G;
MatrixGroup(2, Number Field with defining polynomial
[ x^2 + 1, x^2 + x + 1 ] over the Rational Field)
Generators:
    [                 -1                    0]
    [-i - 2*omega - 1                       1]

    [         1 omega + 1]
    [         0     omega]
```

Complex root data

Let D be the ring of integers of a number field F and assume that F admits a well-defined operation of complex conjugation.

Let L and L^* be free D-modules of rank n in duality via a pairing $L \times L^* \to D : (a, \varphi) \mapsto \langle a, \phi \rangle$. Let $\mu(D)$ be the group of roots of unity in D and let $V = F \otimes_D L$.

A *complex root datum* is a 4-tuple (L, L^*, Φ, ρ), where Φ is a finite subset of L and ρ is a map from Φ to L^* such that, for all $a \in \Phi$:

1. for all $\lambda \in F$, we have $\lambda a \in \Phi$ if and only if $\lambda \in \mu(D)$;
2. for all $\lambda \in \mu(D)$, we have $\rho(\lambda a) = \bar{\lambda}\rho(a)$;
3. $f(a) = 1 - \langle a, \rho(a) \rangle \in \mu(D) \setminus \{1\}$;
4. the reflection r_a of V defined by $r_a(v) = v - \langle v, \rho(a) \rangle a$ and the reflection r_a^* of V^* defined by $r_a^*(\varphi) = \varphi - \langle a, \varphi \rangle \rho(a)$ satisfy:
   - $r_a(\Phi) \subseteq \Phi$ and $r_a^*(\Phi^*) \subseteq \Phi^*$, where $\Phi^* = \rho(\Phi)$.
   - $f(r_b(a)) = f(a)$, for all $a, b \in \Phi$.

The complex Weyl group

The *Weyl group* of a complex root datum $\Sigma = (L, L^*, \Phi, \rho)$ is the group $W = W(\Sigma)$ generated by the reflections $\{ r_a \mid a \in \Phi \}$.

The group W is finite and acts faithfully on Φ. Furthermore, $w(a^*) = w(a)^*$ and $wr_aw^{-1} = r_{w(a)}$, where $a^* = \rho(a)$ and the action of W on V^* is given by $w(\varphi) = \varphi w^{-1}$.

```
>  roots, coroots, rho, W, J := ComplexRootDatum(27);
>  K := BaseRing(J);
>  V := VectorSpace(K,Nrows(J),J);
>  roots := ChangeUniverse(roots,V);
>  coroots := ChangeUniverse(coroots,V);
>  rho := map< roots->coroots | a :-> rho(a) >;
>  R := {@ W!PseudoReflection(a,rho(a)) : a in roots @};
> R[1];
[      -1         0         0]
[       1         1         0]
[-z^4 - z         0         1]
```

Complex Cartan matrices

Suppose that $K \subseteq \mathbb{C}$ is a field and that A and B are $\ell \times n$ matrices over K, where $\ell \geqslant n$.

Let $a_1, a_2, \ldots, a_\ell$ be the rows of A and let $b_1, b_1, \ldots, b_\ell$ be the rows of B.

For $1 \leqslant i \leqslant \ell$ we have a pseudo-reflection r_i of $V = \mathbb{C}^n$ defined by

$$vr_i = v - vb_i^T a_i.$$

The pseudo-reflections $r_1, r_2, \ldots, r_\ell$ are reflections of finite order if and only if the diagonal elements of $I - AB^T$ are roots of unity $\neq 1$. In this case we say that $C = AB^T$ is a *complex Cartan matrix*.

(Note that if C is an $\ell \times \ell$ matrix of rank n then $C = AB^T$ for some $\ell \times n$ matrices A and B.)

The reflection group $G_{3,3,3}$

```
> A,B,J,gen,ord := ComplexRootMatrices(3,3,3);
> A;
[ 1 -1  0]
[ 0  1 -1]
[ 0  1 -z]
> B;
[     1     -1      0]
[     0      1     -1]
[     0      1  z + 1]
> gen,ord;
-z
6
> BaseRing(J);
Cyclotomic Field of order 3 and degree 2
> r := PseudoReflection(A[3],B[3]);
> r;
[      1       0       0]
[      0       0       z]
[      0  -z - 1       0]
```

The reflection group $G_{27} = J_3^{(5)}$

```
> A,B,J, gen, ord := ComplexRootMatrices(27 : NumFld);
>  // A is the identity matrix
> B;
[              2               -1 omega*tau + tau]
[             -1                2              -1]
[     -omega*tau               -1               2]
> gen, ord;
-omega
6
> r1 := PseudoReflection(A[1],B[1]);
> r1;
[              -1                0                0]
[               1                1                0]
[-omega*tau - tau                0                1]
```

Reflection subgroups and parabolic subgroups

Suppose that G is a finite unitary reflection group acting on V.

A *reflection subgroup* of G is simply a subgroup which is generated by reflections.

The *support* of a reflection subgroup H is the subspace of V spanned by the roots of the reflections in H; the *rank* of H is the dimension of its support.

A subgroup of G is *parabolic* if it is the pointwise stabiliser $G(X)$ of a subspace X of V. By a theorem of Steinberg, every parabolic subgroup is a reflection subgroup.

The *parabolic closure* of a reflection subgroup H is the group $G(\mathrm{Fix}(H))$; it is the smallest parabolic subgroup which contains H.

Simple extensions

What are the reflection subgroups of a unitary reflection group and which ones are parabolic?

Theorem

Suppose that H is a reflection subgroup of the finite unitary reflection group G and suppose that $K = \langle H, r \rangle$, where r is a reflection. If K is parabolic and the rank of K is greater than the rank of H, then H is parabolic.

Corollary

If the rank of G is n and if $R = \{r_1, \ldots, r_n\}$ is a set of n reflections which generate G, then every subset of R generates a parabolic subgroup.

Note. Not every reflection group of rank n can be generated by n reflections; some require $n+1$ reflections.

Reflection subgroups of $G_{24} = J_3^{(4)}$



Magma code

```
> orbit_reps := function(G,T)
>   reps := [];
>   while #T gt 0 do
>     t := Rep(T);
>     Append(~reps,t);
>     S := t^G;
>     T := { x : x in T | x notin S };
>   end while;
>   return reps;
> end function;
>
> extendGrp := function(W, refreps, H)
>   N := Normaliser(W,H);
>   X := [ r : r in refreps | r notin H ];
>   orbreps := orbit_reps(N,X);
>   return [ G : r in orbreps |
>         not exists{ E : E in Self() | IsConjugate(W,E,G) }
>                 where G is sub< W | H, r > ];
> end function;
```